HP 35S Calculator Programs

Dr. D.G. Simpson
Department of Physical Sciences and Engineering
Prince George's Community College

May 3, 2014

Programs in this appendix are written for the Hewlett-Packard HP 35s scientific calculators, and can be easily modified to run on other HP calculators that use HP RPN.

Contents

1 Projectile Problem

This program solves the projectile problem: given a target sitting on a hill at coordinates (x_t, y_t) and a cannon with muzzle velocity v_0, at what angle θ should the cannon be aimed to hit the target? The solution is found numerically using Newton's method.

To run the program, enter:

v_0 `ENTER` x_t `ENTER` y_t `ENTER` θ_0 `XEQ J` `ENTER`

Here v_0, x_t, and y_t may be in any consistent set of units, and the angle θ_0 (the initial estimate of the answer) is in degrees. The program returns the angle θ needed to hit the target in degrees.

After running the program, the calculator will be set to degrees mode.

Program Listing

```
J001   LBL J
J002   RAD
J003   →RAD
J004   STO T
J005   R↓
J006   STO Y
J007   R↓
J008   STO X
J009   R↓
J010   STO V
J011   1.015
J012   STO N
J013   RCL T
J014   2
J015   ×
J016   SIN
J017   RCL X
J018   ×
J019   RCL T
J020   COS
J021   x^2
J022   RCL Y
J023   ×
J024   2
J025   ×
J026   −
J027   RCL X
J028   RCL V
J029   ÷
J030   x^2
```

```
J031   9.8
J032   ×
J033   −
J034   RCL T
J035   2
J036   ×
J037   COS
J038   RCL X
J039   ×
J040   2
J041   ×
J042   RCL T
J043   2
J044   ×
J045   SIN
J046   RCL Y
J047   ×
J048   2
J049   ×
J050   +
J051   ÷
J052   RCL T
J053   x<>y
J054   −
J055   STO T
J056   ISG N
J057   GTO J014
J058   →DEG
J059   DEG
J060   RTN
```

Program: `LN=194 CK=644F`

Example. Let $v_0 = 30$ m/s, $(x_t, y_t) = (50$ m, 20 m), and $\theta_0 = 30°$. Enter the above program, then type:

```
30   ENTER   50   ENTER   20   ENTER   30   XEQ J   ENTER
```

The program returns $\theta = 41.5357°$.

2 Kepler's Equation

Given the mean anomaly M (in degrees) and the orbit eccentricity e, this program solves Kepler's equation

$$M = E - e \sin E$$

to find the eccentric anomaly E. This is a very simple implementation—it includes no convergence test, and simply solves Kepler's equation by performing 15 iterations of Newton's method.

To run the program, enter:

M ENTER e XEQ K ENTER

where M is in degrees. The program returns the eccentric anomaly E in degrees. After running the program, the calculator will be set to degrees mode.

Program Listing

```
K001   LBL K
K002   STO E
K003   x<>y
K004   →RAD
K005   STO M
K006   STO A
K007   RAD
K008   1.015
K009   STO K
K010   RCL A
K011   RCL M
K012   RCL A
K013   -
K014   RCL A
K015   SIN
K016   RCL E
K017   ×
K018   +
K019   RCL A
K020   COS
K021   RCL E
K022   ×
K023   1
K024   -
K025   ÷
K026   -
K027   STO A
```

```
K028   ISG K
K029   GTO K011
K030   →DEG
K031   DEG
K032   RTN
```

Program: `LN=102 CK=8EDA`

Example. Let $M = 60°$, $e = 0.15$. Enter the above program, then type:

`60   ENTER   .15   XEQ K   ENTER`   (HP 35s)

The program returns $E = 67.9667°$.

3 Hyperbolic Kepler's Equation

Given the mean anomaly M (in degrees) and the orbit eccentricity e, this program solves the hyperbolic Kepler equation

$$M = e \sinh F - F$$

to find the variable F. This is a very simple implementation—it includes no convergence test, and simply solves the hyperbolic Kepler equation by performing 15 iterations of Newton's method.

To run the program, enter:

M `ENTER` e `XEQ Y` `ENTER`

where M is in degrees. The program returns the variable F.

Program Listing

```
Y001   LBL Y
Y002   STO E
Y003   x<>y
Y004   →RAD
Y005   STO M
Y006   STO A
Y007   1.015
Y008   STO K
Y009   RCL A
Y010   RCL M
Y011   RCL A
Y012   +
Y013   RCL A
Y014   SINH
Y015   RCL E
Y016   ×
Y017   -
Y018   RCL A
Y019   COSH
Y020   RCL E
Y021   ×
Y022   1
Y023   x<>y
Y024   -
Y025   ÷
Y026   -
Y027   STO A
Y028   ISG K
```

```
Y029   GTO Y010
Y030   RTN
```

Program: `LN=96 CK=D135`

Example. Let $M = 60°$, $e = 1.15$. Enter the above program, then type:

```
60   ENTER   1.15   XEQ Y   ENTER
```

The program returns $F = 1.5555$.

4 Barker's Equation

Given the constant $K = \sqrt{GM/(2q^3)}(t - T_p)$, this program solves Barker's equation

$$\tan\left(\frac{f}{2}\right) + \frac{1}{3}\tan^3\left(\frac{f}{2}\right) = \sqrt{\frac{GM}{2q^3}}\,(t - T_p)$$

to find the true anomaly f.

To run the program, enter the dimensionless number

$$K = \sqrt{\frac{GM}{2q^3}}\,(t - T_p)$$

as follows:

K `XEQ B` `ENTER` (HP 35s)

The program returns the anomaly f.

The program will work in either Degrees or Radians mode.

Program Listing

```
B001   LBL B
B002   STO K
B003   ABS
B004   1.5
B005   ×
B006   ENTER
B007   ENTER
B008   ×
B009   1
B010   +
B011   √x
B012   +
B013   3
B014   1/x
B015   y^x
B016   ENTER
B017   ENTER
B018   ×
B019   1
B020   -
B021   x<>y
B022   ÷
B023   RCL K
```

```
B024   ENTER
B025   ABS
B026   ÷
B027   ×
B028   ATAN
B029   2
B030   ×
B031   RTN
```

Program: `LN=100 CK=E151`

Example. Let $K = 19.38$ and set the calculator's angle mode to degrees. Enter the above program, then type:

```
19.38   XEQ B   ENTER
```

The program returns $f = 149.0847°$.

5 Reduction of an Angle

This program reduces a given angle to the range $[0, 360°)$ in degrees mode, or $[0, 2\pi)$ in radians mode. It will work correctly whether the calculator is set for degrees or radians mode.

To run the program:

θ `XEQ R` `ENTER`

The program will return the equivalent reduced angle.

Program Listing

```
R001   LBL R
R002   STO T
R003   -1
R004   ACOS
R005   2
R006   ×
R007   STO Z
R008   RCL T
R009   x ≥ 0?
R010   GTO R022
R011   RCL Z
R012   ÷
R013   +/-
R014   IP
R015   1
R016   +
R017   RCL Z
R018   ×
R019   RCL T
R020   +
R021   RTN
R022   RCL Z
R023   x<>y
R024   x < y?
R025   RTN
R026   x<>y
R027   ÷
R028   IP
R029   RCL Z
R030   ×
R031   RCL T
R032   x<>y
R033   -
```

```
R034   RTN
```

Program: `LN=106 CK=3504`

Example. Let $\theta = 5000°$ and set the calculator's angle mode to degrees. Enter the above program, then type:

`5000   XEQ R   ENTER`   (HP 35s)

The program returns 320°.

6 Helmert's Equation

Given the latitude θ (in degrees) and the elevation H (in meters), this program uses Helmert's equation to find the acceleration due to gravity g.

To run the program, enter:

θ ENTER H XEQ H ENTER

where θ is in degrees and H is in meters. The program returns the acceleration due to gravity g in m/s^2.

After running the program, the calculator will be set to degrees mode.

Program Listing

```
H001   LBL H
H002   DEG
H003   x<>y
H004   2
H005   ×
H006   STO G
H007   COS
H008   0.025928
H009   ×
H010   9.80616
H011   x<>y
H012   -
H013   RCL G
H014   COS
H015   x^2
H016   6.9E-5
H017   ×
H018   +
H019   x<>y
H020   3.086E-6
H021   ×
H022   -
H023   RTN
```

Program: LN=99 CK=E0E9

Example. Let $\theta = 38.898°$, $H = 53$ m. Enter the above program, then type:

38.898 ENTER 53 XEQ H ENTER (HP 35s)

The program returns $g = 9.80052\ m/s^2$.

7 Pendulum Period

Given the length L and amplitude θ of a simple plane pendulum, this program finds the period T, using the arithmetic-geometric mean method.

To run the program, enter:

L ENTER θ XEQ P ENTER

where L is in meters and θ is in degrees. The program returns the period T in seconds.

After running the program, the calculator will be set to degrees mode.

Program Listing

```
P001   LBL P
P002   DEG
P003   STO Q
P004   x<>y
P005   STO L
P006   1
P007   RCL Q
P008   2
P009   ÷
P010   COS
P011   +
P012   2
P013   ÷
P014   STO A
P015   RCL Q
P016   2
P017   ÷
P018   COS
P019   √x
P020   STO G
P021   1.010
P022   STO K
P023   RCL A
P024   ENTER
P025   ENTER
P026   RCL G
P027   +
P028   2
P029   ÷
P030   STO A
P031   R↓
P032   RCL G
```

```
P033   ×
P034   √x
P035   STO G
P036   ISG K
P037   GTO P023
P038   RCL L
P039   9.8
P040   ÷
P041   √x
P042   2
P043   ×
P044   π
P045   ×
P046   RCL A
P047   ÷
P048   RTN
```

Program: `LN=158 CK=44B1`

Example. Let $L = 1.2$ m and $\theta = 65°$. Enter the above program, then type:

`1.2   ENTER   65   XEQ P   ENTER`   (HP 35S)

The program returns $T = 2.3898$ sec.

8 1D Perfectly Elastic Collisions

Given the masses m_1 and m_2 of two bodies and their initial velocities v_{1i} and v_{2i}, this program finds the post-collision velocities v_{1f} and v_{2f}, using

$$
\begin{aligned}
v_{1f} &= \left(\frac{m_1 - m_2}{m_1 + m_2}\right) v_{1i} + \left(\frac{2m_2}{m_1 + m_2}\right) v_{2i} \\
v_{2f} &= \left(\frac{2m_1}{m_1 + m_2}\right) v_{1i} + \left(\frac{m_2 - m_1}{m_1 + m_2}\right) v_{2i}
\end{aligned}
$$

To run the program, enter:

m_1 `ENTER` m_2 `ENTER` v_{1i} `ENTER` v_{2i} `XEQ E` `ENTER`

The program will return the post-collision velocities v_{1f} (in the `X` register) and v_{2f} (in the `Y` register), in the same units.

Program Listing

```
E001   LBL E
E002   STO W
E003   R↓
E004   STO V
E005   R↓
E006   STO N
E007   R↓
E008   STO M
E009   RCL N
E010   −
E011   RCL M
E012   RCL N
E013   +
E014   STO Z
E015   ÷
E016   RCL V
E017   ×
E018   2
E019   RCL N
E020   ×
E021   RCL W
E022   ×
E023   RCL Z
E024   ÷
E025   +
E026   STO X
E027   2
```

```
E028   RCL M
E029   ×
E030   RCL V
E031   ×
E032   RCL Z
E033   ÷
E034   RCL N
E035   RCL M
E036   −
E037   RCL Z
E038   ÷
E039   RCL W
E040   ×
E041   +
E042   RCL X
E043   RTN
```

Program: `LN=131 CK=57A9`

Example. Let $m_1 = 2.0$ kg, $m_2 = 7.0$ kg, $v_{1i} = 4.0$ m/s, and $v_{2i} = -5.0$ kg. Enter the above program, then type:

2 `ENTER` 7 `ENTER` 4 `ENTER` 5 `+/-` `XEQ E` `ENTER`

The program returns $v_{1f} = -10$ m/s in the `X` register, and $v_{2f} = -1$ m/s in the `Y` register.